```
*=====================================================================;
*                                                                     ;
* MODEL OF THE REACTION CH3Cl + OH ---> CH2Cl + H2O AS STUDIED        ;
* IN THE OSLO REACTION CHAMBER                                        ;
*                                                                     ;
* OH radicals are generated by photolysis of O3 in the presence of H2. ;
*                                                                     ;
* Background reactions are taken from oslo_bckgr_reaction.fac         ;
*                                                                     ;
* Written by SRS 15.02.2005                                           ;
*                                                                     ;
*=====================================================================;

EXECUTE OPEN 8 "M:\Fac_jobs\ch3cl_oh.out";

*----- EXPERIMENTAL PARAMETERS ----------------------------------------;
*                                                                     ;
* PATM      Pressure in atm                                           ;
* T         Temperature in K                                          ;
* XH2O      Volume fraction water in PPM                              ;
* XO2       Volume fraction oxygen                                    ;
* XN2       Volume fraction nitrogen                                  ;
* XCl2      Volume fraction Cl2 in PPM                                ;
* XBr2      Volume fraction Br2 in PPM                                ;
* XF2       Volume fraction F2 in PPM                                 ;
* XNO       Volume fraction NO in ppp                                 ;
* XNO2      Volume fraction NO2 in PPB                                ;
* XCH4      Volume fraction CH4 in PPM                                ;
* XN2O5     Volume fraction N2O5 in PPM                               ;
* XO3       Volume fraction O3 in PPM                                 ;
* UVA = 1.0 Blacklamp, 1=On/0=Off                                     ;
* UVB = 0.0 Fluorescent UVB lamp, 1=On/0=Off                          ;
* UVC = 0.0 Fluorescent UVC lamp, 1=On/0=Off                          ;
*;
PARAMETER
PATM      1.0
T       298.0
XH2    5000.0
XH2O      5.0
XO2       0.2
XN2       0.8
XCl2      0.0
XBr2      0.0
XF2       0.0
XNO       0.1
XNO2      0.1
XCO       0.1
XCH4      0.1
XN2O5     0.0
XO3     200.0
UVA       0.0
UVB       0.0
UVC       1.0
NCC       2.46E+19
PPM       1.0E-6
PPB       1.0E-9
PPT       1.0E-12
```

```
XCH3Cl    5.0
;
*----- END OF EXPERIMENTAL PARAMETERS ---------------------------------;


*----- PHOTOLYSIS CONSTANTS -------------------------------------------;
*                                                                       ;
* We should enter new photolysis constants when we have them.          ;
* The values of the photolysis constants are given in                  ;
* the block "COMPILE LIGHTON".                                         ;
*;
PARAMETER
JN2O5 JNO3A JNO3B JHONO JNO2 JO3 JCl2 JH2O2 JHCClO
;
*----- END OF PHOTOLYSIS CONSTANTS ------------------------------------;


*----- VARIABLES FOR BACKGROUND AND PHOTOLYSIS REACTIONS --------------;
*;
VARIABLE
M F HF Br HBr F2 Br2 CO HCl H2 H Cl2 Cl O2 O3
O O1D OH HO2 NO NO2 NO3 N2 N2O CO2
HONO N2O5 HNO3 HO2NO2 HNO HNO2
ClO H2O H2O2 PROD
CH4 CH3 CH3O CH2O HCO CH3O2 CH3OOH CH2OH
CH3ONO2 CH3O2NO2 CH3ONO HCOOH
CCl2O OClO ClO3 Cl2O HOCl ClONO2
Cl2O2 ClNO2 ClOO NOCl ClCO BrO CF2O
CH2OOH HO2CH2O H2CO
FO FO2 FNO FNO2 FONO2
O2ClONO2 ClONO ClNO Cl2O3 HCClO
HOBr BrNO2 BrONO2
;
*----- END OF VARIABLES FOR BACKGROUND AND PHOTOLYSIS REACTIONS -------;


*----- VARIABLES RELATED TO DEGRADATION OF CH3Cl ----------------------;
*;
VARIABLE
CH3Cl CH2Cl CH2ClO2 CH2ClO CH2ClOH CH2ClOOH
;
*----- END OF VARIABLES RELATED TO DEGRADATION OF CH3Cl ---------------;


*----- VARIABLES FOR SINKS --------------------------------------------;
*;
VARIABLE
SINK1 SINK2 SINK3
;
*----- END OF VARIABLES FOR SINKS -------------------------------------;

*----- VARIABLES FOR OUTPUT -------------------------------------------;
*                                                                       ;
* The values are calculated in the block "COMPILE PPM".                ;
*;
VARIABLE
HCClO_PPM SINK1_PPM SINK2_PPM OH_PPM CH3Cl_PPM SINK3_PPM Cl_PPM
```

```
;
*----- END OF VARIABLES FOR OUTPUT ----------------------------------;


COMPILE INSTANT;
M        = NCC * PATM                ;
H2       = XH2 * PPM * NCC * PATM    ;
O2       = XO2 * NCC * PATM          ;
N2       = XN2 * NCC * PATM          ;
F2       = XF2 * PPM * NCC * PATM    ;
Cl2      = XCl2 * PPM * NCC * PATM   ;
Br2      = XBr2 * PPM * NCC * PATM   ;
O3       = XO3 * PPM * NCC * PATM    ;
NO       = XNO * PPB * NCC * PATM    ;
NO2      = XNO2 * PPB * NCC * PATM   ;
N2O5     = XN2O5 * PPB * NCC * PATM  ;
H2O      = XH2O * PPM * NCC * PATM   ;
CO       = XCO * PPM * NCC * PATM    ;
CH4      = XCH4 * PPM * NCC * PATM   ;
CH3Cl    = XCH3Cl * PPM * NCC * PATM;
**;


COMPILE LIGHTON;
*;
JN2O5    = 2.5E-6 * 0.0              ;
JNO3A    = 0.0201 * 0.0              ;
JNO3B    = 0.156 * 0.0               ;
JHONO    = 5.0E-4 * 0.0              ;
JNO2     = 1.3E-3 * 0.0              ;
JO3      = 5.0E-6 * UVB + 3.0E-4 * UVC;* Adjusted to fit experiment;
JCl2     = 1.05E-4 * UVA             ;* Adjusted to fit experiment;
JH2O2    = 2.5E-5 * 0.0              ;
JHCClO   = 1.0E-3                    ;
**;


COMPILE LIGHTOFF;
*;
* All photolysis constants put to zero;
*;
JN2O5    = 0. ;
JNO3A    = 0. ;
JNO3B    = 0. ;
JHONO    = 0. ;
JNO2     = 0. ;
JO3      = 0. ;
JCl2     = 0. ;
JH2O2    = 0. ;
JHCClO   = 0. ;
**;


COMPILE EQUATIONS ;
*;
*----- REACTION SCHEME FOR DEGRADATION OF CH3Cl ----------------------;
*;
```

```
* Unless otherwise stated the rate coefficients are;
* taken from JPL 2-25.;
* The refs. marked as "SQUIB" are taken from the NIST Chemical Kinetics;
* Database on the Web, http://kinetics.nist.gov/index.php.;
*;
% 4.8E-13  : Cl + CH3Cl = HCl + CH2Cl + SINK1;
% 3.6E-14  : OH + CH3Cl = H2O + CH2Cl + SINK2;
% 2.9E-12  : O2 + CH2Cl = CH2ClO2    ;* High-pressure limit;
*;
* The self-reaction of the CH2ClO2 radical proceeds mainly via the;
* first reaction. The branching ratio, alpha, for the second reaction;
* is zero (Catoire et al., J. Phys. Chem., 1994, 98, 2889).;
% 3.63E-12 : CH2ClO2 + CH2ClO2 = O2 + CH2ClO + CH2ClO;*2001ATK/BAU1-56;
% 0.0      : CH2ClO2 + CH2ClO2 = O2 + CH2ClOH + HCClO;* alpha = 0.0;
*;
% 4.92E-12 : HO2 + CH2ClO2 = O2  + CH2ClOOH  ;* 2001ATK/BAU1-56;
% 1.3E-11  : HO2 + CH2ClO2 = H2O + HCClO + O2;* 1996WAL/HUR164-173;
*;
% 907      : CH2ClO      = HCO + HCl  ;* 2001WU/CAR1423-1432;
% 8.0E-14  : O2 + CH2ClO = HO2 + HCClO;* 2001WU/CAR1423-1432;
*;
* The degradation of HCClO is included in the background reactions;
*;
*----- END OF REACTION SCHEME FOR DEGRADATION OF CH3Cl ----------------;
*;
*;
*----- PHOTOLYSIS REACTIONS -------------------------------------------;
*;
% JN2O5  : N2O5 = NO3 + NO2;* 2.5E-6:0;
% JNO3A  : NO3  = NO  + O2 ;* 0.0201:0, JPL;
% JNO3B  : NO3  = NO2 + O  ;* 0.156:0,  JPL;
% JHONO  : HONO = NO  + OH ;* 5.0E-4:0, scaled 350 nm to Cl2;
% JNO2   : NO2  = NO  + O  ;* 1.3E-3:0, scaled 350 nm to Cl2;
% JO3    : O3   = O2  + O1D;* 5.0E-6:0, scaled 350 nm to Cl2;
% JCl2   : Cl2  = Cl  + Cl ;* 5.0E-4:0, locked from experiment;
% JH2O2  : H2O2 = OH  + OH ;* 2.5E-5:0, scaled 350 nm to Cl2;
% JHCClO : HCClO = Cl + HCO;
*;
*----- END OF PHOTOLYSIS REACTIONS ------------------------------------;
*;
*;
*----- BACKGROUND REACTIONS -------------------------------------------;
*;
* Ox Reactions - rate coefficients from JPL 2-25;
*;
% 6.0E-34 : O + O2 + M = O3 + M ;
% 8.0E-15 : O + O3     = O2 + O2;
*;
* O(1D) Reactions - rate coefficients from JPL 2-25;
*;
% 4.0E-11       : O1D + O2     = O    + O2    ;
% 1.2E-10       : O1D + O3     = O2   + O2    ;
% 1.2E-10       : O1D + O3     = O    + O + O2;
% 1.1E-10       : O1D + H2     = OH   + H     ;
% 2.2E-10       : O1D + H2O    = OH   + OH    ;
% 2.6E-11       : O1D + N2     = O    + N2    ;
% 3.5E-37       : O1D + N2 + M = N2O  + M     ;
```

```
% 4.9E-11       : O1D + N2O    = N2   + O2    ;
% 6.7E-11       : O1D + N2O    = NO   + NO    ;
% 1.1E-10       : O1D + CO2    = O    + CO2   ;
% 0.75*1.5E-10  : O1D + CH4    = CH3  + OH    ;
% 0.20*1.5E-10  : O1D + CH4    = CH3O + H     ;
% 0.05*1.5E-10  : O1D + CH4    = CH2O + H2    ;
% 0.09*1.5E-10  : O1D + HCl    = O    + HCl   ;
% 0.24*1.5E-10  : O1D + HCl    = H    + ClO   ;
% 0.67*1.5E-10  : O1D + HCl    = OH   + Cl    ;
% 1.4E-10       : O1D + HF     = OH   + F     ;
% 0.20*1.5E-10  : O1D + HBr    = HBr  + O     ;
% 0.045*1.5E-10 : O1D + HBr    = H    + BrO   ;
% 0.80*1.5E-10  : O1D + HBr    = OH   + Br    ;
% 0.25*2.8E-10  : O1D + Cl2    = O    + Cl2   ;
% 0.74*2.8E-10  : O1D + Cl2    = ClO  + Cl    ;
% 3.6E-10       : O1D + CCl2O  = PROD         ;
% 7.4E-11       : O1D + CF2O   = PROD         ;
*;
* HOx Reactions  - rate coefficients from JPL 2-25;
*;
% 3.3E-11      : O   + OH      = O2   + H      ;
% 5.9E-11      : O   + HO2     = OH   + O2     ;
% 1.7E-15      : O   + H2O2    = OH   + HO2    ;
% 5.7E-32      : H   + O2 + M  = HO2  + M      ;
% 2.9E-11      : H   + O3      = OH   + O2     ;
% 0.87*8.1E-11 : H   + HO2     = OH   + OH     ;
% 0.02*8.1E-11 : H   + HO2     = H2O  + O      ;
% 0.09*8.1E-11 : H   + HO2     = H2   + O2     ;
% 7.3E-14      : OH  + O3      = HO2  + O2     ;
% 6.7E-15      : OH  + H2      = H2O  + H      ;
% 1.9E-12      : OH  + OH      = H2O  + O      ;
% 6.9E-31      : OH  + OH + M  = H2O2 + M      ;
% 1.1E-10      : OH  + HO2     = H2O  + O2     ;
% 1.7E-12      : OH  + H2O2    = H2O  + HO2    ;
% 1.9E-15      : HO2 + O3      = OH   + O2 + O2;
% 1.7E-12      : HO2 + HO2     = H2O2 + O2     ;
% 4.9E-32      : HO2 + HO2 + M = H2O2 + O2 + M ;
*;
* NOx Reactions - rate coefficients from JPL 2-25;
*;
% 9.0E-31 : O    + NO  + M = NO2    + M  ;
% 1.0E-11 : O    + NO2     = NO     + O2 ;
% 2.5E-31 : O    + NO2 + M = NO3    + M  ;
% 1.0E-11 : O    + NO3     = O2     + NO2;
% 3.0E-16 : O    + N2O5    = PROD        ;* Upper limit;
% 3.0E-17 : O    + HNO3    = OH     + NO3;* Upper limit;
% 8.6E-16 : O    + HO2NO2  = PROD        ;
% 1.3E-10 : H    + NO2     = OH     + NO ;
% 7.0E-31 : OH   + NO  + M = HONO   + M  ;
% 2.0E-30 : OH   + NO2 + M = HNO3   + M  ;
% 2.2E-11 : OH   + NO3     = HO2    + NO2;* C7;
% 4.5E-12 : OH   + HONO    = H2O    + NO2;
% 1.5E-13 : OH   + HNO3    = H2O    + NO3;
% 4.6E-12 : OH   + HO2NO2  = PROD        ;
% 8.1E-12 : HO2  + NO      = NO2    + OH ;
% 1.8E-31 : HO2  + NO2 + M = HO2NO2 + M  ;
% 5.0E-16 : HO2  + NO2     = HONO   + O2 ;* C13 Upper limit;
```

```
*;
* C14: Branching ratios for this channel varies from 0.57 to 1;
% 3.5E-12 : HO2  + NO3     = OH     + NO2 + O2;
*;
% 1.9E-14 : NO   + O3      = NO2    + O2      ;
% 2.6E-11 : NO   + NO3     = NO2    + NO2     ;
% 3.2E-17 : NO2  + O3      = NO3    + O2      ;
% 1.0E-18 : NO2  + O3      = O2     + O2 + NO ;* NIST 92HJO/NOT;
*;
% 4.5E-14*EXP(-1260/T) : NO2 + NO3 = NO + NO2 + O2;
*;
% 8.29E-2 : N2O5           = NO2  + NO3     ;* 1997ATK/BAU1329-1499;
% 2.0E-30 : NO2  + NO3 + M = N2O5 + M       ;
% 2.3E-16 : NO3  + NO3     = NO2  + NO2 + O2;
% 1.9E-12 : NO3  + HO2     = HNO3 + O2      ;* NIST 92BEC/RAH;
% 5.0E-19 : O3   + HNO2    = O2   + HNO3    ;* Upper limit;
% 2.0E-21 : N2O5 + H2O     = HNO3 + HNO3    ;
% 0.225   : HO2NO2         = HO2  + NO2     ;* NIST 92LIG/COX;
*                                              Could also go to HONO!;
*;
* Reactions of Organic compounds;
*;
% 1.1E-10              : O  + CH3  = PROD    ;
*;
* D4: Possible formation of competing channel leading to H+HCO2;
% 1.6E-13              : O  + CH2O = OH + HCO;
*;
% 1.5E-13*(1+0.6*PATM) : OH + CO = CO2  + H  ;
*;
% 6.3E-15      : OH    + CH4     = CH3     + H2O ;
% 9.0E-12      : OH    + CH2O    = H2O     + HCO ;
% 0.70*7.4E-12 : OH    + CH3OOH  = CH3O2   + H2O ;
% 0.30*7.4E-12 : OH    + CH3OOH  = CH2OOH  + O2  ;
% 4.0E-13      : OH    + HCOOH   = PROD          ;* D16;
% 3.3E-14      : OH    + CH3ONO2 = PROD          ;* D25;
% 5.0E-14      : HO2   + CH2O    = HO2CH2O       ;* D30: Isomerizes to HOCH2OO;
% 5.2E-12      : HO2   + CH3O2   = CH3OOH  + O2  ;
% 4.0E-19      : NO3   + CO      = NO2     + CO2 ;
% 5.8E-16      : NO3   + CH2O    = HNO3    + HCO ;* D36;
% 3.0E-16      : CH3   + O2      = PROD          ;* Upper limit;
% 4.5E-31      : CH3   + O2 + M  = CH3O2   + M   ;
% 2.6E-12      : CH3   + O3      = PROD          ;
% 5.2E-12      : HCO   + O2      = CO      + HO2 ;
% 5.0E-11      : HCO   + HO2     = PROD          ;* NIST 86TSA/HAM;
% 9.1E-12      : CH2OH + O2      = CH2O    + HO2 ;
% 1.9E-15      : CH3O  + O2      = CH2O    + HO2 ;
% 8.0E-12      : CH3O  + NO      = CH2O    + HNO ;*Upper limit;
% 1.4E-29      : CH3O  + NO  + M = CH3ONO  + M   ;
% 2.0E-13      : CH3O  + NO2     = CH2O    + HONO;
% 5.3E-29      : CH3O  + NO2 + M = CH3ONO2 + M   ;
% 1.0E-17      : CH3O2 + O3      = PROD          ;
*;
*D46: CH3O+CH3O+O2 and CH3+HHCO+O2 are formed, k(rel)=26.2e(-1130/T);
% 3.5E-13      : CH3O2 + CH3O2   = PROD          ;
*;
% 7.7E-12      : CH3O2 + NO      = CH3O     + NO2;
% 1.5E-30      : CH3O2 + NO2 + M = CH3O2NO2 + M  ;
```

```
*;
* FOx Reactions;
*;
% 2.7E-11 : O   + FO      = F     + O2    ;
% 5.0E-11 : O   + FO2     = FO    + O2    ;
% 4.4E-33 : F   + O2  + M = FO2   + M     ;
% 1.0E-11 : F   + O3      = FO    + O2    ;
% 2.6E-11 : F   + H2      = HF    + H     ;
% 1.4E-11 : F   + H2O     = HF    + OH    ;
% 1.8E-31 : F   + NO  + M = FNO   + M     ;
% 6.3E-32 : F   + NO2 + M = FNO2  + M     ;
% 2.3E-11 : F   + HNO3    = HF    + NO3   ;
% 6.7E-11 : F   + CH4     = HF    + CH3   ;
*;
*Upper limit. E54: Two possible pathways: F+2O2 AND FO2+O2;
% 1.0E-14 : FO  + O3      = PROD          ;
*;
% 2.2E-11 : FO  + NO      = NO2   + F     ;
% 2.6E-31 : FO  + NO2 + M = FONO2 + M     ;
% 1.0E-11 : FO  + FO      = F     + F + O2;
% 3.4E-16 : FO2 + O3      = PROD          ;* Upper limit;
% 7.5E-13 : FO2 + NO      = FNO   + O2    ;
% 4.0E-14 : FO2 + NO2     = PROD          ;
% 5.1E-16 : FO2 + CO      = PROD          ;* Upper limit;
% 2.0E-16 : FO2 + CH4     = PROD          ;* Upper limit;
*;
* ClOx Reactions;
*;
% 3.8E-11  : O      + ClO      = Cl       + O2 ;
% 1.0E-13  : O      + OClO     = ClO      + O2 ;
% 1.9E-31  : O      + OClO + M = ClO3     + M  ;
% 4.5E-12  : O      + Cl2O     = ClO      + ClO;
% 1.5E-16  : O      + HCl      = OH       + Cl ;
% 1.7E-13  : O      + HOCl     = OH       + ClO;
% 2.0E-13  : O      + ClONO2   = PROD           ;
% 3.0E-19  : O3     + OClO     = PROD           ;
% 1.0E-19  : O3     + Cl2O2    = PROD           ;* Upper limit;
% 6.7E-14  : OH     + Cl2      = HOCl     + Cl ;
% 1.8E-11  : OH     + ClO      = Cl       + HO2;
% 1.3E-12  : OH     + ClO      = HCl      + O2 ;
% 6.8E-12  : OH     + OClO     = HOCl     + O2 ;
% 8.0E-13  : OH     + HCl      = H2O      + Cl ;
% 5.0E-13  : OH     + HOCl     = H2O      + ClO;
% 3.6E-14  : OH     + ClNO2    = HOCl     + NO2;
% 3.9E-13  : OH     + ClONO2   = PROD           ;
% 3.2E-11  : HO2    + Cl       = HCl      + O2 ;
% 9.1E-12  : HO2    + Cl       = OH       + ClO;
% 5.6E-12  : HO2    + ClO      = HOCl     + O2 ;
% 2.0E-21  : H2O    + ClONO2   = PROD           ;* Upper limit;
% 3.4E-13  : NO     + OClO     = NO2      + ClO;
% 2.0E-14  : NO     + Cl2O2    = PROD           ;* Upper limit;
% 1.0E-31  : NO3    + OClO + M = O2ClONO2 + M  ;* Approximate value;
% 5.0E-17  : NO3    + HCl      = HNO3     + Cl ;* Upper limit;
% 1.0E-21  : HO2NO2 + HCl      = PROD           ;* Upper limit;
% 6.64E-11 : Cl     + Cl       = Cl2            ;* NIST 76HIP/TRO;
% 2.7E-33  : Cl     + O2   + M = ClOO     + M  ;
% 1.2E-11  : Cl     + O3       = ClO      + O2 ;
```

```
% 1.6E-14  : Cl     + H2       = HCl      + H + SINK3 ;
% 4.1E-13  : Cl     + H2O2     = HCl      + HO2;
% 1.0E-11  : Cl     + HONO     = HCl      + NO2;* Assumption!;
% 9.0E-32  : Cl     + NO   + M = NOCl     + M  ;
% 1.3E-30  : Cl     + NO2  + M = ClONO    + M  ;
% 1.8E-31  : Cl     + NO2  + M = ClNO2    + M  ;
% 2.4E-11  : Cl     + NO3      = ClO      + NO2;
*;
*% 1. : Cl+N2O=ClO+N2;*See F56 tab 1, 1e-17 at 1000 K, Ea=43kcal/mol.;
*;
% 2.0E-16  : Cl    + HNO3     = PROD      ;* Upper limit;
% 1.0E-13  : Cl    + HO2NO2   = PROD      ;* Upper limit;
% 1.3E-33  : Cl    + CO   + M = ClCO + M  ;
% 1.0E-13  : Cl    + CH4      = HCl  + CH3;
% 7.3E-11  : Cl    + CH2O     = HCl  + HCO;
% 2.0E-13  : Cl    + HCOOH    = PROD      ;
*;
*F63: Two pathways with unknown branching ratios: ClO+CH3O and HCl+CH2O2;
% 1.6E-10  : Cl    + CH3O2    = PROD      ;
*;
% 5.7E-11  : Cl    + CH3OOH   = PROD      ;
% 2.3E-13  : Cl    + CH3ONO2  = PROD      ;
% 5.8E-11  : Cl    + OClO     = ClO + ClO ;
% 2.3E-10  : Cl    + ClOO     = Cl2 + O2  ;
% 1.2E-11  : Cl    + ClOO     = ClO + ClO ;
% 9.6E-11  : Cl    + Cl2O     = Cl2 + ClO ;
% 1.0E-10  : Cl    + Cl2O2    = PROD      ;
*;
*F83: Two pathways with unknown branching ratios: Cl2+OH and HCl+ClO;
% 1.6E-12  : Cl    + HOCl     = PROD          ;
*;
% 8.1E-11  : Cl    + ClNO     = NO     + Cl2  ;
% 1.0E-11  : Cl    + ClONO2   = Cl2    + NO3  ;* F85: More than 95%;
% 1.4E-17  : ClO   + O3       = ClOO   + O2   ;* Upper limit;
% 1.0E-18  : ClO   + O3       = OClO   + O2   ;* Upper limit;
% 1.0E-19  : ClO   + H2       = PROD          ;* Upper limit;
% 1.7E-11  : ClO   + NO       = NO2    + Cl   ;
% 1.8E-31  : ClO   + NO2  + M = ClONO2 + M    ;
% 4.7E-13  : ClO   + NO3      = ClOO   + NO2  ;
% 6.0E-19  : ClO   + N2O      = PROD          ;* Upper limit;
% 4.0E-18  : ClO   + CO       = PROD          ;* Upper limit;
% 4.0E-18  : ClO   + CH4      = PROD          ;* Upper limit;
% 1.0E-15  : ClO   + H2CO     = PROD          ;* Upper limit;
*;
*F117: Two pathways with unknown branching ratios: ClOO+CH3O and CH3OCl+O2;
% 2.2E-12  : ClO   + CH3O2    = PROD          ;
*;
% 4.8E-15  : ClO   + ClO      = Cl2   + O2    ;
% 8.0E-15  : ClO   + ClO      = ClOO  + Cl    ;
% 3.5E-15  : ClO   + ClO      = OClO  + Cl    ;
% 1.6E-32  : ClO   + ClO  + M = Cl2O2 + M     ;
% 6.2E-32  : ClO   + OClO + M = Cl2O3 + M     ;
% 1.0E-20  : HCl   + ClONO2   = PROD          ;* Upper limit;
% 6.92E-12 : HCO   + Cl2      = HCClO + Cl    ;* NIST 88TIM;
% 6.0E-5   : HCClO            = PROD          ;* Estimated wall loss
% 7.63E-13 : HCClO + Cl       = ClCO  + HCl   ;* NIST 90LIB/ZAB;
% 5.0E-13  : HCClO + OH       = ClCO  + H2O   ;* 2001ATK/BAU1-56;
```

```
% 1.0E+3   : ClCO            = Cl    + CO    ;* Estimate;
% 6.49E-13 : ClOO  + M       = Cl    + O2 + M;* 2001ATK/BAU1-56;
* Very important to include this reaction!;
*;
* BrOx Reactions;
*;
% 4.1E-11      : O   + BrO     = Br    + O2 ;
% 3.8E-14      : O   + HBr     = OH    + Br ;
% 2.8E-11      : O   + HOBr    = OH    + BrO;
% 4.2E-11      : OH  + Br2     = HOBr  + Br ;
% 7.5E-11      : OH  + BrO     = PROD       ;
% 1.1E-11      : OH  + HBr     = H2O   + Br ;
% 2.0E-12      : HO2 + Br      = HBr   + O2 ;
% 2.1E-11      : HO2 + BrO     = PROD       ;
% 1.0E-16      : NO3 + HBr     = HNO3  + Br ;* Upper limit;
% 1.2E-12      : Br  + O3      = BrO   + O2 ;
% 5.0E-16      : Br  + H2O2    = HBr   + HO2;* Upper limit;
% 4.2E-31      : Br  + NO2 + M = BrNO2 + M  ;
% 1.6E-11      : Br  + NO3     = BrO   + NO2;
% 1.1E-12      : Br  + H2CO    = HBr   + HCO;
% 3.4E-13      : Br  + OClO    = BrO   + ClO;
*;
* G37: Upper limit for production of OBrO+O2 is about 2e-18;
% 2.0E-17      : BrO + O3      = PROD            ;
*;
% 2.1E-11      : BrO + NO      = NO2    + Br     ;
% 5.2E-31      : BrO + NO2 + M = BrONO2 + M      ;
% 1.0E-12      : BrO + NO3     = PROD            ;
% 0.85*3.2E-12 : BrO + BrO     = Br     + Br + O2;* 298K;
% 0.15*3.2E-12 : BrO + BrO     = Br2    + O2     ;* 298K;
*;
*----- END OF BACKGROUND REACTIONS ----------------------------------;
*;
**;


SETPSTREAM 1 8 ;
TIME OH_PPM SINK1_PPM SINK2_PPM HCClO_PPM CH3Cl_PPM SINK3_PPM Cl_PPM;
**;


COMPILE OUT ;
PSTREAM 1 ;
**;


COMPILE PPM;
OH_PPM    = OH / (NCC * PPM);
HO2_PPM   = HO2 / (NCC * PPM);
SINK1_PPM = SINK1 / (NCC * PPM);
SINK2_PPM = SINK2 / (NCC * PPM);
SINK3_PPM = SINK3 / (NCC * PPM);
HCClO_PPM = HCClO / (NCC * PPM);
CH3Cl_PPM = CH3Cl / (NCC * PPM);
Cl_PPM    = Cl / (NCC * PPM);
CALL OUT;
**;
```

```
WHENEVER TIME = 1050 * (+10) 0  % CALL PPM;
         TIME =     0 CALL LIGHTON  RESTART;
         TIME =   300 CALL LIGHTOFF RESTART;
         TIME =  1500 CALL LIGHTON  RESTART;
         TIME =  1800 CALL LIGHTOFF RESTART;
         TIME =  3000 CALL LIGHTON  RESTART;
         TIME =  3300 CALL LIGHTOFF RESTART;
         TIME =  4500 CALL LIGHTON  RESTART;
         TIME =  4800 CALL LIGHTOFF RESTART;
         TIME =  6000 CALL LIGHTON  RESTART;
         TIME =  6300 CALL LIGHTOFF RESTART;
         TIME =  7500 CALL LIGHTON  RESTART;
         TIME =  7800 CALL LIGHTOFF RESTART;
         TIME =  9000 CALL LIGHTON  RESTART;
         TIME =  9300 CALL LIGHTOFF RESTART;
         TIME = 10500 CALL LIGHTON  RESTART;
**;

BEGIN;

STOP;
```